Definite Integration in *Mathematica* V3.0

Victor Adamchik

Introduction

The aim of this paper is to provide a short description of definite integration algorithms implemented in *Mathematica* Version 3.0.

```
$Version
```

Linux 3.0 (April 25, 1997)

Proper Integrals

All proper integrals in *Mathematica* are evaluated by means of the Newton-Leibniz theorem

$$\int_a^b f(x)\, dx = F(b) - F(a)$$

where $F(x)$ is an antiderivative. It is well-known that the Newton-Leibniz formula in the given form does not hold any longer if the antiderivative $F(x)$ has singularities on an interval of integration (a, b). Let us consider the following integral

$$\int_0^4 \frac{x^2 + 2x + 4}{x^4 - 7x^2 + 2x + 17}\, dx$$

where the integrand is a smooth integrable function on an interval $(0, 4)$.

```
Plot[(4 + 2 x + x^2)/(17 + 2 x - 7 x^2 + x^4), {x, 0, 4}]
```



- Graphics -

As it follows from the Risch structure theorem the correspondent indefinite integral is doable in elementary functions

```
int = ∫ (x^2 + 2 x + 4)/(x^4 + 2 x - 7 x^2 + 17) dx
```

$$\frac{1}{2}\tan^{-1}\left(\frac{-x-1}{x^2-4}\right)-\frac{1}{2}\tan^{-1}\left(\frac{x+1}{x^2-4}\right)$$

If we simply substitute limits of integration into the antiderivative we get an incorrect result.

```
Limit[int, x → 4] - Limit[int, x → 0]
```

$$-\tan^{-1}\left(\frac{1}{4}\right)-\tan^{-1}\left(\frac{5}{12}\right)$$

This is because the antiderivative is not a continuous function on an interval (0, 4). It has a jump at $x = 2$, which is easy to see in the following graphic.

```
Plot[int, {x, 0, 4}]
```



- Graphics -

The right way of applying the Newton-Leibniz theorem is to take into account an influence of the jump

```
Limit[int, x -> 4, Direction ->  1] -
Limit[int, x -> 2, Direction -> -1] +
Limit[int, x -> 2, Direction ->  1] -
Limit[int, x -> 0, Direction -> -1]
```

$$\pi - \tan^{-1}\left(\frac{1}{4}\right) - \tan^{-1}\left(\frac{5}{12}\right)$$

Mathematica evaluates definite integrals in precisely that way.

$$\int_0^4 \frac{x^2 + 2x + 4}{x^4 - 7x^2 + 2x + 17}\, dx$$

$$\pi - \tan^{-1}\left(\frac{1}{4}\right) - \tan^{-1}\left(\frac{5}{12}\right)$$

The origin of discontinuities of antiderivatives along the path of integration is not in the method of indefinite integration but rather in the integrand. In the discussed example, the integrand has four singular poles that become branch points for the antiderivative.

```
NRoots[x^4 - 7 x^2 + 2 x + 17 == 0, x]
```

$$x == -1.95334 - 0.244028\, i \lor x == -1.95334 + 0.244028\, i \lor$$
$$x == 1.95334 - 0.755972\, i \lor x == 1.95334 + 0.755972\, i$$

Connected in pairs these points make two branch cuts. And the path of integration crosses one of them. The following `ContourPlot` clearly exposes the problem.

```
ContourPlot[Evaluate[Re[int /. x -> x + I y]], {x, -3, 4}, {y, -3, 4},
 ContourShading -> False, Contours -> 20, PlotPoints -> 40,
 Epilog -> {Hue[0], Thickness[0.005], Line[{{0, 0}, {4, 0}}]}]
```



- ContourGraphics -

We see that in a complex plane of the variable *x* the antiderivative has two branch cuts (bold black vertical lines) and the path of integration, the line (0,4), intersects the right branch cut. Obviously, by varying the constant of integration we can change the form of the antiderivative so that we would get various forms of branch cuts. Here we understand the constant of integration as a function f(x) such that $\frac{df(x)}{dx}$ is zero. As a simple example let us consider the step-wise constant function $\frac{\sqrt{x^2}}{x}$

$$\text{Simplify}\left[\partial_x \frac{\sqrt{x^2}}{x}\right]$$

0

Thinking hard, we can built an antiderivative that does not have a branch cut crossing a given interval of integration. However, we can never get rid of branch cuts!

Analysis of the singularities of antiderivatives is a time consuming and sometimes heuristic process, especially if trigonometric or special functions are involved in antiderivatives. In the latter case **Integrate** may not be able to detect all singular points on the interval of integration, which will result in a warning message

— *Integrate::gener : Unable to check convergence*

You should pay attention to the message since it warns you that the result of the integration might be wrong.

Improper Integrals

It is quite clear that the above procedure cannot cover the whole variety of definite integrals. There are two reasons behind that. First, the correspondent indefinite integral cannot be expressed in finite terms of functions represented in *Mathematica*. For instance,

```
∫Cos[Sin[x]] ⅆx
```

$$\int \cos(\sin(x))\, dx$$

However, the definite integral with the specific limits of integration is doable.

```
∫_0^π Cos[Sin[x]] ⅆx
```

$$\pi J_0(1)$$

Second, even if an indefinite integral can be done, it requires a great deal of effort to find limits at the end points. Here is an example,

```
∫_0^(π/2) Tan[x]^(1/π) ⅆx
```

$$\frac{1}{2}\pi \sec\left(\frac{1}{2}\right)$$

This is an improper integral since the top limit is a singular point of the integrand. The result of indefinite integration is

```
∫Tan[x]^(1/π) ⅆx
```

$$\frac{\pi\, {}_2F_1\left(\frac{1+\pi}{2\pi}, 1; 1+\frac{1+\pi}{2\pi}; -\tan^2(x)\right)\tan^{1+\frac{1}{\pi}}(x)}{1+\pi}$$

To find the limit at $x = \frac{\pi}{2}$ one has to, first, develop the asymptotic expansion of the hypergeometric Gauss function at ∞, and second, construct the asymptotic scale for the Puiseux series. Currently, *Mathematica*'s **`Series`** structure is based on power series that allow only rational exponents.

Following we present a short description of the algorithm for evaluating improper integrals. The overall idea has been given in [1] and [3]. Some practical details regarding "logarithmic" cases are described in [2] and [4].

■ The General Idea

By means of the Mellin integral transform and Parseval's equality, a given improper integral is transformed to a contour integral over the straight line $(\gamma - i\infty,\ \gamma + i\infty)$ in a complex plane of the parameter z:

$$\int_0^\infty f_1(x)\, f_2\!\left(\frac{z}{x}\right) \frac{dx}{x} = \frac{1}{2\pi i} \int_{\gamma - i\infty}^{\gamma + i\infty} f_1{}^*(s)\, f_2{}^*(s)\, z^{-s}\, ds \qquad (1)$$

where $f_1^*(x)$ and $f_2^*(x)$ are Mellin transforms of $f_1(x)$ and $f_2(x)$

$$f^*(s) = \int_0^\infty f(x)\, x^{s-1}\, dx$$

The real parameter γ in the formula (1) is defined by conditions of existence of Mellin transforms $f_1^*(x)$ and $f_2^*(x)$. Finally, the residue theorem is used to evaluate the contour integral in the right side of the formula (1)

$$\frac{1}{2\pi i} \oint_\Gamma f(z)\, dz = \sum_{k=1}^{m} \operatorname*{res}_{z = a_k} f(z)$$

where Γ is a closed contour, and a_k are poles of *f(z)* that lie in a domain bounded by Γ. The success of this scheme depends on two factors: first, the Mellin image of an integrand $f_1(x)\, f_2(\frac{z}{x})$ must exist, and, second, it should be represented in terms of Gamma functions. If these conditions are satisfied then the contour integral in (1), called the Mellin-Barnes integral or the Meijer G-function, can almost always be expressed in finite terms of hypergeometric functions. This fact is known as Slater's theorem (see [1] and [5]). We said "almost always", since there is a special case of the G-function when the latter cannot be reduced to hypergeometric functions but to their derivatives with respect to parameters. By analogy with linear differential equations with polynomial coefficients, such a singular case of the G-function is named a logarithmic case. The modified Bessel function $K_0(z)$ is such a classical example, since its series representation

$$K_0(z) = -I_0(z) \log\!\left(\frac{z}{2}\right) + \sum_{k=0}^{\infty} \frac{\psi(k+1)}{k!^2} \left(\frac{z^2}{4}\right)^k$$

cannot be expressed via hypergeometrics.

■ Mellin-Barnes Integrals

These integrals are defined by (see[6])

$$\frac{1}{2\pi i} \oint_{\mathcal{L}} g(s)\, z^{-s}\, ds$$

where

$$g(s) = \frac{\prod_{j=1}^{n} \Gamma(a_j + s) \prod_{j=1}^{m} \Gamma(b_j - s)}{\prod_{j=1}^{k} \Gamma(c_j + s) \prod_{j=1}^{l} \Gamma(d_j - s)}$$

and the contour $\mathcal{L}$ is a line that separates poles of $\Gamma(a_j + s)$ from $\Gamma(b_j - s)$. Let us investigate when the integral exists. On a straight line $\mathcal{L}$, $= (\gamma - i\infty,\ \gamma + i\infty)$ the real part of $s \in \mathcal{L}$ is bounded, and $|\mathrm{Im}(s)| \to \infty$. Using the Stirling asymptotic formula

$$|\Gamma(x + i\,y)| = \sqrt{2\pi}\ |y|^{x-\frac{1}{2}}\ e^{-\frac{\pi}{2}|y|}\left(1 + O\left(\frac{1}{|y|}\right)\right),\ |y| \to \infty$$

we can deduce that the integrand $g(s)\,z^{-s}$ vanishes exponentially as $|\mathrm{Im}(s)| \to \infty$ if $m + n - k - l > 0$, and $|\arg(z)| < \frac{\pi}{2}\,(m + n - k - l)$. If $m + n - k - l = 0$, then z must be real and positive. Some additional conditions are required here (see details in [1]).

It is clear that we can not simply imply the residue theorem to this contour integral. We need initially to transform the contour $\mathcal{L}$ to the closed one. There are two possibilities: we can either transform $\mathcal{L}$ into the left loop $\mathcal{L}_{-\infty}$, the contour encircling all poles of $\Gamma(a_j + s)$, or to the right loop $\mathcal{L}_{+\infty}$, the contour encircling the poles of $\Gamma(b_j - s)$. The criteria of which contour should be chosen appears from the convergence of the integral along that contour. On the left-hand loop $\mathcal{L}_{-\infty}$, the imaginary part of $s \in \mathcal{L}_{-\infty}$ is bounded and $\mathrm{Re}(s) \to -\infty$. Assuming that $|\arg(-s)| < \frac{\pi}{2}$, and making use of the Stirling formula

$$\Gamma(z) = \sqrt{2\pi}\ z^{z-\frac{1}{2}}\ e^{-z}\left(1 + O\left(\frac{1}{z}\right)\right),\ z \to \infty,\ |\arg(z)| < \pi$$

and the reflection formula of the Gamma function, we find that the Mellin-Barnes integral over the loop $\mathcal{L}_{-\infty}$ exists, if $n + l - m - k > 0$. If $n + l - m - k = 0$, then z must be within the unit disk $|z| < 1$. If z is on a unit circle $|z| = 1$, the integral converges if

$$\mathrm{Re}\left(\sum_{j=1}^{n} a_j + \sum_{j=1}^{m} b_j + \sum_{j=1}^{k} c_j + \sum_{j=1}^{l} d_j\right) < -k + n - 1$$

On the right-hand loop $\mathcal{L}_{+\infty}$, the imaginary part of $s \in \mathcal{L}_{+\infty}$ is bounded and $\mathrm{Re}(s) \to +\infty$. Proceeding similarly to the above, we find that the Mellin-Barnes integral over the loop $\mathcal{L}_{+\infty}$ exists, if $n + l - m - k < 0$. If $n + l - m - k = 0$, then z must be outside of the unit disk. Additional conditions are required if z is on a unit circle.

After determining the correct contour, the next step is to calculate residues of the integrand. Since the integrand contains only Gamma functions, this task is more or less formal. We don't even need to calculate residues of the integrand, but go straightforwardly to the generalized hypergeometric functions. The only obstacle is the logarithmic case, which occurs when the integrand has multiple poles. We have to separate two subclasses here: the integrand that has a finite number of multiple poles, and the integrand that has infinitely many multiple poles. The *Mathematica* integration routine has a full implementation of the former case. In the latter, the integration artificially restricted by the second order poles, since for the higher order poles it would lead to infinite sums with higher order polygamma functions. This class of infinite sums are extremely hard to deal with, symbolically and numerically. If such a situation is detected **`Integrate`** returns the Meijer G-function. However, there are some very special transformations of the G-function, which could avoid bulky infinite sums with polygamma functions, and give a nice result in terms of known functions. In [4] I demonstrated a few transformations that reduce the order of the G-function and make it possible to handle special class of Bessel integrals in terms of Bessel functions.

■ An Example

Consider the integral

$$\Omega = \int_0^\infty \frac{\sin(x)}{x\,(x^2+1)}\, dx$$

According to the formula (1), we have

$$\Omega = \int_0^\infty \frac{1}{x^2+1}\sin\left(\frac{1}{1/x}\right)\frac{dx}{x} = \frac{1}{2\pi i}\int_{\gamma-i\infty}^{\gamma+i\infty} f_1^*(s)\, f_2^*(s)\, ds$$

where $f_1^*(s)$ and $f_2^*(s)$ are Mellin transforms of e^{-x} and $\sin(\frac{1}{x})$:

```
f1*[s] = ∫_0^∞ x^(s-1)/(x^2 + 1) dx
```

$$\text{If}\left(\text{Re}(s) > 0 \land \text{Re}(s) < 2,\ \frac{1}{2}\pi\csc\left(\frac{\pi s}{2}\right),\ \int_0^\infty \frac{x^{s-1}}{x^2+1}\, dx\right)$$

```
f2*[s] = ∫_0^∞ Sin[1/x] x^(s-1) dx
```

$$\text{If}\left(\text{Re}(s) > -1 \land \text{Re}(s) < 1,\ -\Gamma(-s)\sin\left(\frac{\pi s}{2}\right),\ \int_0^\infty x^{s-1}\sin\left(\frac{1}{x}\right) dx\right)$$

Therefore,

$$\Omega = -\frac{\pi}{4\pi i}\int_{\gamma-i\infty}^{\gamma+i\infty} \Gamma(-s)\, ds$$

where γ is defined by conditions of the existence of Mellin transforms

$$0 < \gamma = \text{Re}(s) < 1$$

As it follows from the previous section we can transform the integration contour $(\gamma-\infty, \gamma+i\infty)$ into the right loop $\mathcal{L}_{+\infty}$. Then, using the residue theorem we evaluate the integral as a sum of residues at simple poles $s = 1, 2, \ldots$.

```
Ω = -π/2 Sum[(-1)^k/k!, {k, 1, ∞}]
```

$$-\frac{(1-e)\pi}{2e}$$

■ Meijer G-Function

We did not intend to give a complete picture of the Meijer G-function here, but only necessary facts. In Version 3.0 the G-function is defined by

$$\text{MeijerG}[\{\{a_1, a_2, \ldots, a_n\}, \{a_{n+1}, a_{n+2}, \ldots, a_p\}\}, \{\{b_1, b_2, \ldots, b_m\}, \{b_{m+1}, b_{m+2}, \ldots, b_q\}\}, z] =$$

$$G_{p,q}^{m,n}\left(z \,\middle|\, \begin{matrix} a_1, & a_2, & \ldots, & a_p \\ b_1, & b_2, & \ldots, & b_q \end{matrix}\right) =$$

$$\frac{1}{2\pi i} \oint_{\mathcal{L}} \frac{\prod_{i=1}^{m} \Gamma(b_i + s) \prod_{i=1}^{n} \Gamma(1 - a_i - s)}{\prod_{i=n+1}^{p} \Gamma(a_i + s) \prod_{i=m+1}^{q} \Gamma(1 - b_i - s)} z^{-s} \, ds$$

and the contour $\mathcal{L}$ is a left (or right) loop $\mathcal{L}_{-\infty}$ (or $\mathcal{L}_{+\infty}$) separated poles of $\Gamma(b_j + s)$ from $\Gamma(1 - a_j - s)$. The current implementation of the Meijer function has two noticeable features that differ it from the classical G-function. First, you are not allowed to choose or move the contour of integration, and, second, **`MeijerG`** has an optional parameter (the classical G-function does not) that regulates the branch cut. The Meijer function is supported symbolically and numerically.

```
MeijerG[{{0}, {}}, {{0}, {}}, z]
```

$$\frac{1}{z+1}$$

```
MeijerG[{{-1}, {1, 2}}, {{0, 1/2}, {}}, 1/2]
```

$$\text{MeijerG}\left(\{\{-1\}, \{1, 2\}\}, \left\{\left\{0, \frac{1}{2}\right\}, \{\}\right\}, \frac{1}{2}\right)$$

```
N[%]
```

0.209893

Only in very trivial cases **`MeijerG`** is simplified automatically to the lower level special functions. Beyond that all further transformations are assigned to **`FunctionExpand`**:

```
FunctionExpand[%%]
```

$$\sqrt{\pi} - \frac{\sqrt{\pi}\, I_0(1)}{e} - \frac{2\sqrt{\pi}\, I_1(1)}{e}$$

`MeijerG` is interlaced with indefinite and definite integration, and with solving linear differential equations with polynomial coefficients. Here is an example related to definite integration:

```
Integrate[x^2 BesselK[0, x]/Sqrt[x^2 - 1], {x, 1, ∞}]
```

$$\frac{1}{4}\sqrt{\pi}\,\text{MeijerG}\left(\left\{\{\},\left\{-\frac{1}{2}\right\}\right\},\{\{-1,0,0\},\{\}\},\frac{1}{4}\right)$$

The first element of the second argument of **MeijerG**, which is {-1,0,0}, indicates that the integrand of the correspondent contour integral contains the product of three Gamma functions $\Gamma(s-1)\,\Gamma(s)\,\Gamma(s)$ and has an infinite number of triple poles at s=-k, k=0,1,2, ... From the design point of view it is definitely an advantage for **Integrate** to return a short object, **MeijerG**, rather than enormous infinite sums involving derivatives of the Gamma function.

■ Hypergeometric Functions

The essential part of integration is the generalized hypergeometric function, which is defined by

$$\text{HypergeometricPFQ}[\{a_1,\ \ldots,\ a_p\},\ \{b_1,\ \ldots,\ b_q\},\ z] = {}_pF_q\left(z \,\middle|\, \begin{matrix} a_1, & a_2, & \ldots, & a_p \\ b_1, & b_2, & \ldots, & b_q \end{matrix}\right) = \sum_{k=0}^{\infty} \frac{\prod_{j=1}^{p}(a_j)_k}{\prod_{j=1}^{q}(b_j)_k}\,\frac{z^k}{k!}$$

where $(a)_k$ is a Pochhammer symbol

$$(a_j)_k = \prod_{l=1}^{k}(a_j + l - 1) = \frac{\Gamma(a_j + k)}{\Gamma(a_j)}$$

Conditions of convergence can be easily obtained by applying the d'Alembert test. It follows, ${}_pF_q$ converges for all finite z if $p \le q$, and for $|z| < 1$ if $p = q + 1$. Additional conditions are required on the circle of convergence $|z|=1$. The above series definition of ${}_pF_q$ has an exceptional case, when $p = 2$ and $q = 0$ - such **HypergeometricPFQ** is defined via the confluent hypergeometric function **HypergeometricU**

```
HypergeometricPFQ[{a, b}, {}, z]
```

$$\left(-\frac{1}{z}\right)^a U\left(a,\, a-b+1,\, -\frac{1}{z}\right)$$

For $|z| \geqslant 1$ and $p = q + 1$, the hypergeometric function is defined as an analytic continuation via the Mellin-Barnes integral, with the branch cut $[1, \infty)$:

$$_{q+1}F_q\left(z \,\middle|\, \begin{matrix} a_1, & a_2, & \ldots, & a_{q+1} \\ b_1, & b_2, & \ldots, & b_q \end{matrix}\right) = \frac{\prod_{k=1}^{q}\Gamma(b_k)}{\prod_{k=1}^{q+1}\Gamma(a_k)}\,\frac{1}{2\pi i}\int_{\gamma-i\infty}^{\gamma+i\infty}\frac{\Gamma(s)\ \prod_{j=1}^{q+1}\Gamma(a_j - s)}{\prod_{j=1}^{q}\Gamma(b_j - s)}\,(-z)^{-s}\,ds$$

where the straight line $(\gamma - i\infty, \gamma + i\infty)$ separates poles of $\Gamma(s)$ from $\Gamma(a_j - s)$.

Almost all symbolic simplifications of ${}_p F_q$ to the lower level functions are done automatically

Hypergeometric2F1$\left[\frac{1}{3}, \frac{1}{3}, \frac{4}{3}, \frac{1}{2}\right]$

$$\frac{1}{9} \sqrt[3]{2} \left(\sqrt{3}\, \pi + 3 \log(2)\right)$$

HypergeometricPFQ$\left[\left\{1, \frac{1}{3}, \frac{1}{3}\right\}, \left\{\frac{4}{3}, \frac{4}{3}\right\}, 1\right]$

$$\frac{1}{9}\, \psi^{(1)}\left(\frac{1}{3}\right)$$

```
HypergeometricPFQ[{1, 1, 1, 1}, {4, 4, 4}, -1]
```

$$\frac{27}{4} (-169 + 96 \log(4) + 30\, \zeta(3))$$

The exception is the Gauss function **`Hypergeometric2F1`**. Since it has so many different transformation and simplification rules some of them are assigned to **`FunctionExpand`**:

Hypergeometric2F1$\left[1, \frac{1}{4}, \frac{5}{4}, \frac{1}{4}\right]$

$${}_2F_1\left(1, \frac{1}{4}; \frac{5}{4}; \frac{1}{4}\right)$$

```
FunctionExpand[%]
```

$$\frac{\cot^{-1}\left(\sqrt{2}\right)}{\sqrt{2}} - \frac{\log\left(1 - \frac{1}{\sqrt{2}}\right)}{2\sqrt{2}} + \frac{\log\left(1 + \frac{1}{\sqrt{2}}\right)}{2\sqrt{2}}$$

Principal-Value Integrals

Consider the integral

$$\int_{-1}^{2} \frac{1}{x}\, dx$$

— *Integrate::idiv : Integral of* $\frac{1}{x}$ *does not converge on* $\{-1, 2\}$.

$$\int_{-1}^{2} \frac{1}{x}\, dx$$

which does not exist in the Riemann sense, since it has a nonintegrable singularity at $x = 0$. However, if we isolate $x = 0$ by $\epsilon_1 > 0$ from the left and by $\epsilon_2 > 0$ from the right and take a limit of correspondent integrals when $\epsilon_1 \to 0$ and $\epsilon_2 \to 0$, we obtain

$$\lim_{\substack{\epsilon_1 \to 0 \\ \epsilon_2 \to 0}} \left(\int_{-1}^{-\epsilon_1} \frac{1}{x} \, dx + \int_{\epsilon_2}^{2} \frac{1}{x} \, dx \right) = \lim_{\substack{\epsilon_1 \to 0 \\ \epsilon_2 \to 0}} (\log(2) + \log(\epsilon_1) - \log(\epsilon_2))$$

The double limit exists, and so it is a given integral, if and only if $\epsilon_1 = \epsilon_2$. Such understanding of divergent integrals is called the principal-value or the Cauchy principal-value. It is easy to see that if an integral exists in the Riemann sense, it exists in the Cauchy sense. Thus, the class of Cauchy integrals is larger than the class of Riemann integrals.

In Version 3.0, definite integrals in the Riemann sense and principal-value integrals are separated by the new option **`PrincipalValue`**. If you want to evaluate an integral in the Cauchy sense, set the option **`PrincipalValue`** to **`True`** (the default setting is **`False`**). Here is an example

```
Integrate[1/x, {x, -1, 2}, PrincipalValue -> True]
```

$\log(2)$

```
Integrate[1/Sinh[5 x + 1], {x, -1, 1}, PrincipalValue -> True]
```

$\frac{1}{5} \log(\coth(2) \tanh(3))$

```
Integrate[x/Sin[3 x], {x, -1/2 (3 π), 3 π/2}, PrincipalValue -> True]
```

$\frac{4 C}{9}$

If the integrand contains high-order polynomials, **`Integrate`** returns **`RootSum`** objects

```
Integrate[1/(x^7 + x + 1), {x, -2, 1}, PrincipalValue -> True]
```

$$\frac{i \pi}{1 + 7 \, \mathrm{Root}(\#1^7 + \#1 + 1 \,\&, 1)^6} - \mathrm{RootSum}\left(\#1^7 + \#1 + 1 \,\&, \frac{\log(-\#1 - 2)}{7 \#1^6 + 1} \,\&\right) + \mathrm{RootSum}\left(\#1^7 + \#1 + 1 \,\&, \frac{\log(1 - \#1)}{7 \#1^6 + 1} \,\&\right)$$

Here are examples of integrals with movable singularities

$$\texttt{Integrate}\left[\frac{1}{\texttt{a - Tan[x]}^2}, \left\{\texttt{x, 0, } \frac{\pi}{2}\right\}, \texttt{PrincipalValue} \to \texttt{True}\right]$$

$$\text{If}\left(a > 0, \frac{\pi}{2a+2}, \int_0^{\frac{\pi}{2}} \frac{1}{a - \tan^2(x)} \, dx\right)$$

$$\texttt{Integrate}\left[\frac{\texttt{x}^\lambda}{\texttt{x - a}}, \{\texttt{x, 0, } \infty\}, \texttt{PrincipalValue} \to \texttt{True}\right]$$

$$\text{If}\left(a > 0 \land \text{Re}(\lambda) > -1 \land \text{Re}(\lambda) < 0, -a^\lambda \pi \cot(\pi \lambda), \int_0^\infty \frac{x^\lambda}{x - a} \, dx\right)$$

Here **`Integrate`** detects that the integrand has a singular point along the path of integration if the parameter *a* is real positive.

New Classes of Integrals

In this section we give a short overview of specific classes of definite integrals, which were essentially improved in the new version. *Mathematica* now is able to calculate almost all indefinite and about half of definite integrals from the well-known collection of integrals compiled by Gradshteyn and Ryzhik. Moreover, the Version 3.0 makes it possible to calculate thousands of new integrals not included in any published handbooks.

■ integrals of rational functions

The **`RootSum`** object has been linked to **`Integrate`** to display the result of integration in a more elegant and shorter way

$$\int_0^1 \frac{\texttt{x - 1}}{\texttt{x}^7 + \texttt{x}^3 + 1} \, d\texttt{x}$$

$$\text{RootSum}\left(\#1^7 + \#1^3 + 1 \,\&, \frac{\log(1 - \#1)\,\#1 - \log(1 - \#1)}{7\,\#1^6 + 3\,\#1^2} \,\&\right) - \text{RootSum}\left(\#1^7 + \#1^3 + 1 \,\&, \frac{\log(-\#1)\,\#1 - \log(-\#1)}{7\,\#1^6 + 3\,\#1^2} \,\&\right)$$

■ logarithmic and Polylogarithm integrals

$$\int_0^1 \frac{\texttt{x}^2 \, \texttt{Log[x] Log[x + 1]}}{\texttt{x + 1}} \, d\texttt{x}$$

$$\frac{1}{8}\left(-20 + \pi^2 + 16 \log(2) - \zeta(3)\right)$$

$$\int_0^1 \frac{\text{Log}[1-x]^3}{(1-x\,z)^2}\, dx$$

$$-\frac{6\,\text{Li}_3\left(\frac{z}{z-1}\right)}{(z-1)\,z}$$

■ elliptic integrals

The package, conducted an elliptic integration, is now autoloaded (as are as all other integration packages), so you don't need to worry about its loading anymore

$$\int_{-\frac{\pi}{4}}^{\frac{\pi}{4}} \frac{1}{\sqrt{\text{Cos}[x]}}\, dx$$

$$4\,F\left(\frac{\pi}{8}\,\middle|\,2\right)$$

$$\int_0^{\frac{\pi}{2}} \frac{x}{\sqrt{\text{Sin}[x]}}\, dx$$

$${}_3F_2\left(\frac{1}{2}, \frac{3}{4}, 1; \frac{3}{2}, \frac{3}{2}; 1\right)$$

■ integrals involving Bessel functions

$$\int_0^\infty t\,\text{Exp}[-t^2]\,\text{BesselI}[2, t]\,\text{BesselK}[2, t]\, dt$$

$$-3 + \frac{1}{4}\sqrt{e}\,K_2\left(\frac{1}{2}\right)$$

$$\int_0^\infty \frac{\text{BesselJ}[1, x]\,\text{Exp}[-\frac{2}{x}]}{x}\, dx$$

$$2\,J_1(2)\,K_1(2)$$

$$\int_0^\infty \text{AiryAi}[x]^2\, dx$$

$$-\frac{\Gamma\left(-\frac{1}{6}\right)}{12\,\sqrt[3]{2}\,\sqrt[6]{3}\,\pi^{3/2}}$$

■ integrals involving non-analytic functions

$$\int_{-1-I}^{2+2\,I} \mathbf{Sin[Abs[x]]}\, dx$$

$$(3+3\,i)\left(\frac{\sqrt{2}}{3}-\frac{\cos(\sqrt{2})}{3\sqrt{2}}-\frac{\cos(2\sqrt{2})}{3\sqrt{2}}\right)$$

$$\int_{0}^{\frac{3\pi}{2}} \mathbf{Max[Sin[x],\ Cos[2\,x]]}\, dx$$

$$\frac{3\sqrt{3}}{2}$$

Additional Features

Mathematica's capability for definite integration gained substantial power in the new version. Other essential features of **`Integrate`** not discussed in the previous sections are convergence tests and the assumptions mechanism.

■ Conditions

In Version 3.0 **`Integrate`** is "conditional." In most cases, if the integrand or limits of integration contains symbolic parameters, **`Integrate`** returns an **`If`** statement of the form

If[*conditions*, *answer*, *held integral*]

which gives necessary conditions for the existence of the integral. For example

$$\int_{0}^{\infty} x^{\lambda-1}\, \mathbf{Exp[-\alpha\, x]}\, dx$$

$$\mathrm{If}\left(\mathrm{Re}(\alpha)>0 \land \mathrm{Re}(\lambda)>0,\ \alpha^{-\lambda}\,\Gamma(\lambda),\ \int_0^\infty e^{-x\,\alpha}\, x^{\lambda-1}\, dx\right)$$

Setting the option **`GenerateConditions`** to **`False`** prevents **`Integrate`** from returning conditional results (as in Version 2.2):

```
Integrate[x^(λ-1) Exp[-α x], {x, 0, ∞}, GenerateConditions → False]
```

$$\alpha^{-\lambda}\,\Gamma(\lambda)$$

If a given definite integral has symbolic parameters, then the result of integration essentially always depends on certain specific conditions on those parameters. In this example the restrictions $\mathrm{Re}(\alpha)>0$ and $\mathrm{Re}(\lambda)>0$

came from conditions of the convergence. Even when a definite integral is convergent, some other conditions on parameters might appear. For instance, the presence of singularities on the integration path could lead to essential changes when the parameters vary. The next section is devoted to the convergence of definite integrals.

■ Convergence

The new integration code contains criteria for the convergence of definite integrals. Each time **Integrate** examines the integrand for convergence

$$\int_{-1}^{1} \frac{\text{Cos}[x]}{x}\, dx$$

— *Integrate::idiv : Integral of* $\frac{\cos(x)}{x}$ *does not converge on* $\{-1, 1\}$.

$$\int_{-1}^{1} \frac{\cos(x)}{x}\, dx$$

This integral has a nonintegrable singularity at $x = 0$. Thus, **Integrate** generates a warning message and returns unevaluated. However, the integral exists in the Cauchy sense. Setting the option **PrincipalValue** to **True**, we obtain

$$\text{Integrate}\left[\frac{\text{Cos}[x]}{x}, \{x, -1, 1\}, \text{PrincipalValue} \to \text{True}\right] \text{ // FullSimplify}$$

0

Consider another integral with a symbolic parameter α

$$\int_{0}^{1} x^{\rho-1}\, \text{ArcTan}[x]\, dx$$

$$\text{If}\left(\text{Re}(\rho) > -1,\ \frac{\psi^{(0)}\left(\frac{\rho+1}{4}\right) - \psi^{(0)}\left(\frac{\rho+3}{4}\right) + \pi}{4\,\rho},\ \int_{0}^{1} x^{\rho-1} \tan^{-1}(x)\, dx\right)$$

The integral has a singular point at $x = 0$, which is integrable only if $\text{Re}(\rho) > -2$.

If you are sure that a particular integral is convergent or you don't care about the convergence, you can avoid testing the convergence by setting the option **GenerateConditions** to **False**. It will make **Integrate** return an answer a bit faster.

Setting **GenerateConditions** to **False** also lets you evaluate divergent integrals

$$\text{Integrate}\left[\frac{1}{x}, \{x, 0, 2\}, \text{GenerateConditions} \to \text{False}\right]$$

log(2)

■ Assumptions

The new option **Assumption** is used to specify particular assumptions on parameters in definite integrals. Consider the integral with the arbitrary parameter y

$$\int_0^\infty \frac{e^{-x^2-x\,y}}{\sqrt{x}}\,dx$$

Here we set the option **Assumptions** to $\mathrm{Re}(y) > 0$

```
Integrate[Exp[-x^2 - x y]/Sqrt[x], {x, 0, ∞}, Assumptions → Re[y] > 0]
```

$$\frac{1}{2}\,e^{\frac{y^2}{8}}\,\sqrt{y}\,K_{\frac{1}{4}}\left(\frac{y^2}{8}\right)$$

Setting **Assumptions** to $\mathrm{Re}(y) < 0$, we get a different form of the answer

```
Integrate[Exp[-x^2 - x y]/Sqrt[x], {x, 0, ∞}, Assumptions → Re[y] < 0]
```

$$\frac{e^{\frac{y^2}{8}}\,\pi\,\sqrt{-y}\,\left(I_{-\frac{1}{4}}\left(\frac{y^2}{8}\right)+I_{\frac{1}{4}}\left(\frac{y^2}{8}\right)\right)}{2\sqrt{2}}$$

though the integral is a continuous function with respect to the parameter y

```
Plot[NIntegrate[Exp[-x^2 - x y]/Sqrt[x], {x, 0, ∞}], {y, -2, 2}]
```



- Graphics -

The next integral is discontinuous with respect to the parameter γ

```
Integrate[Cos[x] (1 - Cos[γ x]) / x^2, {x, 0, ∞}, Assumptions → γ > 1]
```

$$\frac{1}{2}\pi(\gamma-1)$$

```
Integrate[Cos[x] (1 - Cos[γ x]) / x^2, {x, 0, ∞}, Assumptions → -1 < γ < 1]
```

0

```
Integrate[Cos[x] (1 - Cos[γ x]) / x^2, {x, 0, ∞}, Assumptions → γ < -1]
```

$$\frac{1}{2}\pi\left(\frac{1}{\sqrt{\frac{1}{\gamma^2}}}-1\right)$$

If the given assumptions exactly match the generated assumptions, then the latter don't show up in the output; otherwise, **`Integrate`** produces assumptions complementary to ones given:

```
Integrate[x^(ν-1) (1 - x)^(μ-1), {x, 0, 1}, Assumptions → Re[ν] > 0]
```

$$\text{If}\left(\text{Re}(\mu)>0,\ \frac{\Gamma(\mu)\,\Gamma(\nu)}{\Gamma(\mu+\nu)},\ \int_0^1 (1-x)^{\mu-1}\,x^{\nu-1}\,dx\right)$$

References

1. O.I. Marichev, *Handbook of integral transforms of higher transcendental functions: theory and algorithmic tables*, Ellis Horwood, 1983.

2. V.S. Adamchik, K.S. Kolbig, A definite integral of a product of two polylogarithms, *SIAM J. Math. Anal.* **4**(1988), 926-938.

3. V.S. Adamchik, O.I. Marichev, The algorithm for calculating integrals of hypergeometric type functions and its realization in REDUCE system, *Proc. Conf. ISSAC'90*, Tokyo (1990), 212-224.

4. V. Adamchik, The evaluation of integrals of Bessel functions via G-function identities, *J. Comp. Applied Math.* **64**(1995), 283 - 290.

5. L.J. Slater, *Generalized Hypergeometric Functions*, Cambridge Univ. Press, 1966.

6. Y.L. Luke, *The Special Functions and Their Approximations*, Vol.1, Academic Press, 1969.